
Introduction


Microsoft’s Windows 95 Game SDK was designed to support the development of 2D and 3D games on the PC platform. One component of this SDK, the DirectPlay component, is intended to facilitate multi-player networked games. Of course, there is nothing in the SDK that requires that you use it for game purposes. DirectPlay itself is application neutral. Can we use it to provide a useful Smalltalk capability?





Consider the following scenario. A friend is developing an application and encounters a problem. He calls me at home asking for help. I log on to my machine, download his classes, and then proceed to look at the code. As I browse his methods, I may ask him questions about its design, how a method works, etc. Of course, he may have difficulty responding if he has to do it all from memory. More likely, he’ll start up his own Smalltalk environment and try to peruse the methods I am asking him questions about. 





Rather than keep synchronized via negotiated voice interaction, it would be much easier if we could just synchronize our two browsers. Since conventional browsers can’t do that, we developed our own net enabled browser which we can start up in VisualAge as follows:





	NetEnabledBrowser new open





Before the browser actually appears, we first need to choose a specific DirectPlay service provider that will be responsible for inter-application communication. On our pentium, three providers were available (see the prompter of Figure 1). We chose the TCP provider. A subsequent prompter (see Figure 2) allows us to specify whether to start a new session (a session is a network group) or to connect to an existing session.  For the scenario envisaged above, we would start a new session (by clicking on Yes to become a server); our friend and any one else who might wish to connect would not (they would click on No to become clients). A server is subsequently prompted for user identification (see Figure 3). In our case, we started a session called “Smalltalk Help Group.” A network enabled browser then appears; it looks the same as a standard class hierarchy browser.





By contrast, clients are first provided with a list of existing sessions to choose from (see the prompter of Figure 4), then a user identification prompter appears (Figure 3), and finally a network enable browser appears.





DirectPlay provides facilities for targeting messages to specific users or for broadcasting. In our application, we simply broadcast all new method selections. So a method selected by anyone will be seen by everyone else in the group. There is good documentation for DirectPlay in the SDK. However, other sources such as Dr Dobbs [1] might be better to provide insight on how to use the facilities.


The Overall Design


DirectPlay requires users to make use of relatively standard callback facilities (standard for Microsoft) for enumerating list of providers, lists of sessions, lists of players, etc. It also provides a handful of special data structures that must be initialized, interrogated, and modified. It is possible to deal with all these low-level issues directly in Smalltalk but it is much easier to do it in C. So we decided to provide a higher-level interface to the DirectDraw API in C which we compiled using VisualC++ to build a DLL. We then implemented a DirectPlay class in Smalltalk to use the DLL API. The NetEnabledBrowser was created as a subclass of the class hierarchy browser with facilities to communicate over the net using a DirectPlay object. 


The C Software


A small portion of the C software supporting the send/receive functions is shown in Listing 1. Note the little details that must be attended to when receiving a message. First: is there a message? Second: if there is a message, is there sufficient space to retrieve it (if there wasn’t, a retry is need). Fourth: we eliminate messages to ourselves (which actually bounce back when we broadcast message).





The complete API that we developed on top of the existing Microsoft DirectPlay API consists of the following functions:





	void openDirectPlay (char *aDoubleByteGuidString);


	void closeDirectPlay ();


	char *providerList ();


	char *sessionList (long timeout);


	char *playerList ();


	void chooseProvider (long index);


	void chooseSession (long index);


	void choosePlayer (long index);


	void server (char *formalName, char *friendlyName);


	void client (char *formalName, char *friendlyName);


	char *lastError ();


	Message *receive ();


	void send (Message *stuff);





The Message structure is described in Listing 1. As a rule, Smalltalk strings maintain a hidden length component in the string object whereas C strings are terminated with the '\0' character (they are said to be null terminated). So explicit conversions are needed. We can do that as follows:





	aSmalltalkString nullTerminated 


	(OSStringZ address: aCStringPointer) asString





A pointer retrieved from a C DLL is just an integer to Smalltalk. In general, C structures can be created in Smalltalk as a subclass of OSObject. OSStringZ is a subclass specifically designed for null terminated strings. Instances can specifically reach into a memory location and create an equivalent Smalltalk string by sending it the asString message. 





Byte arrays can be passed unchanged to C but more work is needed to get one back. The size must be explicitly handled independently in both directions. That’s why we created the Message structure in C; it can be viewed as a length-prefixed byte array. We probably should have created a corresponding OSMessage class but we didn’t bother. The following shows how we might extract a byte array from C (a bit messy):





	aByteArray := ByteArray new: desiredSize.


	(OSObject address: aByteArrayPointer offset: 0)


		memcpyFrom: 0 to: aByteArray  size - 1 


		into: aByteArray startingAt: 0.





To provide access to these functions, we placed the DLL, which in our case is called “netEnabler.dll”, in the same directory as all the other Smalltalk DLLs. Then we can write methods such as the following in any class:





	privateChoosePlayer: index


		<c: void 'netEnabler' :choosePlayer int32>


		^self primitiveFailed





	privateServer: formalName and: friendlyName


	<c: void 'netEnabler' :server struct struct>


	^self primitiveFailed





	privateClient: formalName and: friendlyName


		<c: void 'netEnabler' :client struct struct>


		^self primitiveFailed





	privateErrorString


		<c: pointer 'netEnabler' :lastError>





Note that int32 indicates a long (32 bit integer) to be passed, struct is a string passed in, pointer is a string or byte array returned. 





All windows applications need to provide a global unique identifier (guid) for use by direct play. The service provider also have their own guids which can be retrieved via enumeration facilities and they must be provided when the provider is activated. We managed to make all that transparent to users of our DLL but we couldn’t do it for the application guid. A guid is manufactured by a special Microsoft program “uuidgen” which we found in the c:\msdev\bin directory. There is a special windows function “CLSIDFromString” that can convert strings generated by “uuidgen” to the required internal format. For a long time, we couldn’t get it to work for us. It turned out that the actual id data had to be surrounded by braces even though all the online windows documentation said that square brackets had to be used. Additionally, it wouldn’t work using normal strings. We had to use double byte strings (16 bit characters). Fortunately, we could convert a standard string into a double byte string easily:





		an8BitByteString asDBString





Now consider the list of function provided earlier. Functions open… and close… are straightforward. Functions providerList, sessionList, and playerList return one C string with special character '\1' separating individual substrings. In Smalltalk, '\1' is “1 asCharacter”. We can easily convert such a string to a collection of substrings by executing


	


		aString substrings: 1 asCharacter





A provider list is a list of service provider names, a session list is a list of existing session names, and a player list is a even length list of successive formal and friendly names. This information can be easily submitted to application users via Smalltalk list prompters. The index into the list can be used to choose one by providing it as a parameter to one of the corresponding functions chooseProvider, chooseSession, and choosePlayer. Choosing a provider, for example, causes the corresponding provider to be activated; the actual information needed to do this is the guid of the provider which is maintained in private internal data structures accessible from the index. Choosing a session (or player) is similar; given an index, a session (or player) id is retrieved from session (player) data structures. As a rule, these ids are not in any particular order, so a simple mapping is needed. We actually store away all the information supplied by the enumeration facilities but only the above is actually required for the applications we have in mind so far. In general, the indices start at 1 except for the player indices. Special index 0 is permitted and it corresponds to “all players”. To send a message to a specific player, we first “choose” that player and then send a message. To broadcast (which is the default until another player is chosen), we simply “choose” player 0.





In general, any function can fail for one reason or another. If it does fail, lastError returns a non-empty string; otherwise, it returns an empty string. Function send causes bytes of data to be sent to the “chosen” player. Function receive retrieve the bytes from whoever sent it.





When starting up, we have to supply a formal name and a friendly name by indicating whether or not we wish to be a server (there is only one) or a client. In a game environment, the server might do the heavy-duty work with the clients doing the easier local-work. The server also starts first and finishes last. The name of the session is the server’s formal name. For our network enable browser, servers and clients do the same work.


The DirectPlay Class


The DirectPlay class (see Listing 2) interfaces with the C functions in the DLL via corresponding private messages of the form private…; e.g., privateProviderList. Analogous public messages (without the private prefix) provide access using Smalltalk convenient representations. For example, privateProviderList provides a C string where the individual provider names are separated by character '\1'; providerList (oops, we actually called it networkProviders), by contrast, returns an ordered collection of strings (provider names).





For use with the net enabled browser, we use messages sendStrings: and receiveStrings which are actually implemented in terms of lower-level messages that can read and write into Smalltalk C-like structures (instances of subclasses of OSObject). We could have implemented the sendStrings: and receiveStrings messages directly in a much simpler way but we anticipated future applications that might need the more general mechanism. The only real test that it works is the use we make of it to extract bytes for an OSByteArray.


The Net Enabled Browser


The net enabled browser (list Listing 3) is a subclass of EtClassesBrowser. We were thinking of calling our class dspNetEnabledBrowser (for “damn stupid prefix” net enabled browser) to follow IBM conventions but decided against it. Our browsers keep track of one direct play object and a polling process. Currently, direct play is not event oriented. It has no mechanism to inform users of new messages when they come in. We actually have to ask it for the next message. To do this conveniently, we create a special process (see methods connectPollingProcess and disconnectPollingProcess) that polls every so often (see class method pollingRateInMilliseconds to see how often) for a direct play message. When it gets one (see method receive), the message in variable strings is decoded and performed. Performing a message is easy because it has the following simple form.





	#(oneParameterSelectorString string1 string2 string3 …)





We simply execute





	self 


		perform: strings first asSymbol 


		with: strings





Currently, we support only two messages:





	#(directPlayHello:)


	#(directPlaySelect: classNameString selector)





where classNameString might be 'Account' or 'Account class' depending on whether the method selector is referring to an instance or class method. See





For simplicity, we don’t send messages to specific players. Rather we broadcast all messages. When new clients join a session, they send the first message. Only the server (see method directPlayHello:) reacts to this message and it reacts by executing informDirectPlayOfMSelection which broadcasts a message of the second form resulting in the recipients (see method directPlaySelect:) updating their browser.





When any player in the session selects a new method, the browser ultimately sends itself the changedSelectedMethod message which we override to execute informDirectPlayOfSelection. Hence everyone in the group can control what is seen by the others.





One problem that we encountered (which may be a bug on our part) was the receipt of a garbage message on startup. We filter that out in method directPlayGarbage:. It didn’t seem to happen after that point, however.


Conclusion


In this column, we showed how we might develop a browser that could be used for interactive collaborative browsing. There are a number of directions we might pursue. Currently, modifications to methods are not visible because we don’t send the changed text. So we can’t use it for remote programming. It doesn’t seem to be too difficult to add these extensions.





We also just received a beta of new direct play software from Microsoft. It supports automatic server migration, event triggering, and shared data spaces. We’ll probably need substantial reworking to use the new release.

Where to Obtain the Code


Source code for this article can be obtained on the World Wide Web at http://www.objectpeople.on.ca/software/joop. It is implemented in VisualAge and provided both as source and as an Envy export file.

 Figure 1  The network provider prompter.

 Figure 2  Starting a new session (Yes) or connecting to an existing session (No).

 Figure 3  The user identification prompter.

 Figure 4  The session choice prompter.





Listing 1 A small sampling of the C code used to create the 'netEnabler' DLL.





#define WIN32_LEAN_AND_MEAN





#include <stdio.h>


#include <dplay.h>





char *errorMessage; 


char *lastError () {


	return errorMessage;


}





void noteError (char *message) {


	errorMessage = message;


}





#define run(code,message) \


	{HRESULT result = code; \


	if (result != DP_OK) { \


		noteError (message); \


	}}





DPID userIdentifier;





struct Message {


	unsigned long size; //size of data (not of Message).


	BYTE data [1];


};





Message emptyReceiveBuffer = {0, 0};


Message *receiveBuffer = NULL; 


long receiveBufferSizeLimit = 0; //maximum value for size





void reallocateReceiveBuffer (long newSize) {


	//Allocate Message where data length is rounded up to a 


	//multiple of an 'unsigned long'.


	if (receiveBuffer != NULL) 


		delete [] (unsigned long *) receiveBuffer;


	long typeSize = sizeof (unsigned long);


	long longSize = newSize / typeSize +


		1 /*to round up data bytes*/ + 1 /*for size*/;


	receiveBuffer = (Message *) new unsigned long [longSize];


	if (receiveBuffer == NULL) {


		receiveBufferSizeLimit = 0;


		noteError ("Not enough space for receive buffer");


	}


	receiveBufferSizeLimit = longSize * typeSize - typeSize;


}





Message *receive () {


	//It is the user's responsibility to copy the data since 


	//subsequent messages overwrite the same space.





	//Is there a message?


	unsigned long count;


	run (directPlay->GetMessageCount (userIdentifier, &count), 


		"Unknown player attempting to receive");


	if (count == 0) {


		//No, there is no message.


		return &emptyReceiveBuffer;


	} 





	//Try to get the message.


	unsigned long byteSize = receiveBufferSizeLimit;


	DPID fromIdentifier; DPID toIdentifier = userIdentifier;


	HRESULT result = directPlay->Receive (&fromIdentifier, 


		&toIdentifier, DPRECEIVE_TOPLAYER, 


		&receiveBuffer->data, &byteSize);


	if (result == DPERR_BUFFERTOOSMALL) {


		//No, need more space.


		reallocateReceiveBuffer (byteSize);


		byteSize = receiveBufferSizeLimit;


		//Try again, should work this time.


		result = directPlay->Receive (&fromIdentifier, 


			&toIdentifier, DPRECEIVE_TOPLAYER, 


			&receiveBuffer->data, &byteSize);


	}


	if (result != DP_OK) {


		noteError ("Receive replied with an error");


		return &emptyReceiveBuffer;


	}


	if (fromIdentifier == userIdentifier) 


		return receive (); //Ignore self messages.


	receiveBuffer->size = byteSize;


	return receiveBuffer;


}





Listing 2 The DirectPlay class.





class:									DirectPlay


superclass:							Object


instance variables:			logonType





class methods





timing control


timeoutInMilliseconds


	^5000





instance methods





interacting


open: guidString


	self privateOpenDirectPlay: guidString nullTerminated


close


	^self privateCloseDirectPlay





networkProviders


	"Providers are collections of strings."


	^self asStrings: (self asString: self privateProviderList)


networkSessions


	"Sessions are collections of strings."


	^self asStrings: (self asString: (self privateSessionList: 


		self class timeoutInMilliseconds)) 


players


	"Players are collections of 'formal, friendly name' strings."


	| strings pairs |


	strings := self 


		asStrings: (self asString: self privatePlayerList).


	pairs := OrderedCollection new.


	1 to: strings size // 2 by: 2 do: [:index |


		pairs add: (strings at: index), ', ', (strings at: index+1)].


	^pairs





chooseNetworkProvider: index


	^self privateChooseProvider: index


choosePlayer: index


	^self privateChoosePlayer: index


chooseSession: index


	^self privateChooseSession: index





receiveStrings


	^self asStrings: self receiveByteArray


receiveByteArray


	^(self receiveAs: OSByteArray) asBytes


receiveAs: OSclass


	"Receive into a fixed or variable size structure."


	| pointer |


	pointer := self privateReceive.


	OSclass variableSize = 0


		ifTrue: [


			^self unpack: pointer intoFixedStructure: OSclass]


		ifFalse: [


			^self unpack: pointer intoVariableStructure: OSclass]





sendStrings: aCollection


	"Packed format: length (4 bytes), actual data (length bytes)."


	| stream bytes |


	stream := WriteStream on: (String new: 400).


	stream nextPutAll: '    '. "room for length"


	aCollection do: [:string | 


		stream nextPutAll: string; nextPut: self stringSeparator].


	stream skip: -1.


	bytes := stream contents asByteArray.	


	bytes int32At: 0 put: bytes size - 4.


	self privateSend: bytes


sendByteArray: aByteArray


	self sendAs: aByteArray


sendAs: anOSObject


	| bytes |


	bytes := self packIntoByteArray: anOSObject


	self privateSend: bytes





logonAsClientNamed:  formalName and: friendlyName


	logonType := #client.


	self 


		privateClient: formalName nullTerminated 


		and: friendlyName nullTerminated





logonAsServerNamed:  formalName and: friendlyName


	logonType := #server.


	self 


		privateServer: formalName nullTerminated


		and: friendlyName nullTerminated





errorString


	^self asString: self privateErrorString





logonType


	"Returns #server or #client"


	^logonType





support for interacting 


packIntoByteArray: OSObjectOrByteArray


	"Packed format: length (4 bytes), actual data (length bytes)."


	| size bytes |


	size := OSObjectOrByteArray size. 


	bytes := ByteArray new: size + 4.


	bytes int32At: 0 put: size.


	OSObjectOrByteArray 


		memcpyFrom: 0 to: size - 1 into: bytes startingAt: 4.


	^bytes





unpack: pointer intoFixedStructure: OSclass


	"Unpack into a fixed size structure. 


	Packed format: length (4 bytes), actual data (length bytes)."


	| actualByteSize requiredByteSize |


	actualByteSize := (OSObject address: pointer) int32At: 0.


	requiredByteSize :=  OSclass fixedSize.


	self 


		validateFixedSizes: actualByteSize 


		and: requiredByteSize.


	"NB: Return a copy; direct play reuses its message buffers."


	^(OSclass address: pointer offset: 4) copy





unpack: pointer intoVariableStructure: OSclass


	"Unpack into a variable size structure. 


	Packed format: length (4 bytes), actual data (length bytes)."


	| structure actualByteSize requiredByteSize count |


	actualByteSize := (OSObject address: pointer) int32At: 0.


	count := (actualByteSize - OSclass fixedSize) 


		// OSclass variableSize.


	requiredByteSize := OSclass fixedSize + 


		(OSclass variableSize * count).


	self 


		validateVariableSizes: actualByteSize 


		and: requiredByteSize.


	structure := OSclass newVariable: count.


	(OSObject address: pointer offset: 4)


		memcpyFrom: 0 to: actualByteSize - 1 


		into: structure startingAt: 0.


	^structure





validateFixedSizes: actualSize and: requiredSize


	actualSize = requiredSize ifTrue: [^self].


	^self error: 'Received ', actualSize printString, 


		' bytes for conversion to ', requiredSize printString, 


		' byte OS structure'





validateVariableSizes: actualSize and: requiredSize


	actualSize = requiredSize ifTrue: [^self].


	^self error: 'Received ', actualSize printString, 


		' bytes for conversion to OS structure; ',


		(requiredSize - actualSize) printString, 


		' excess bytes dropped.'





relaying to the DLL


privateOpenDirectPlay: guidDoubleByteString


	<c: void 'NetEnabler' :openDirectPlay struct>


	^self primitiveFailed


privateCloseDirectPlay


	<c: void 'netEnabler' :closeDirectPlay>


	^self primitiveFailed


privateProviderList


	<c: pointer 'netEnabler' :providerList>


	^self primitiveFailed


privateSessionList: timeoutInMilliseconds


	<c: pointer 'netEnabler' :sessionList int32>


	^self primitiveFailed


privatePlayerList


	<c: pointer 'netEnabler' :playerList>


	^self primitiveFailed


privateChooseProvider: index


	<c: void 'netEnabler' :chooseProvider int32>


	^self primitiveFailed


privateChooseSession: index


	<c: void 'netEnabler' :chooseSession int32>


	^self primitiveFailed


privateChoosePlayer: index


	<c: void 'netEnabler' :choosePlayer int32>


	^self primitiveFailed


privateServer: formalName and: friendlyName


	<c: void 'netEnabler' :server struct struct>


	^self primitiveFailed


privateClient: formalName and: friendlyName


	<c: void 'netEnabler' :client struct struct>


	^self primitiveFailed


privateErrorString


	<c: pointer 'netEnabler' :lastError>


	^self primitiveFailed


privateReceive


	"Receives a pointer to a byte array encoded as follows: 


	size of other data (4 bytes), other data (size bytes)."


	<c: pointer 'netEnabler' :receive>


	^self primitiveFailed


privateSend: byteArray


	"Sends a byte array encoded as follows: 


	size of other data (4 bytes), other data (size bytes)."


	<c: void 'netEnabler' :send pointer>


	^self primitiveFailed





utilities


asString: pointer


	"Converts a DLL string pointer into a Smalltalk string."


	^(OSStringZ address: pointer) asString





asStrings: aStringOrByteArray


	"Converts a string into a collection of strings."


	aStringOrByteArray isEmpty 


		ifTrue: [^#()]


		ifFalse: [


			^aStringOrByteArray asString 


				subStrings: self stringSeparator]





stringSeparator


	^1 asCharacter








Listing 3 The NetEnabledBrowser class.





class:									NetEnabledBrowser


superclass:							EtClassesBrowser 


instance variables:			directPlayObject pollingProcess


class variables:					ActiveNetEnabledBrowser 





class methods 





timing control


pollingRateInMilliseconds


	^2000





when under duress


reset


	ActiveNetEnabledBrowser := nil





instance methods





opening/closing


open


	ActiveNetEnabledBrowser notNil ifTrue: [


		^ActiveNetEnabledBrowser bringToFront].


	ActiveNetEnabledBrowser := self.


	self connectToNetwork.


	self directPlayInactive ifTrue: [


		self messagePrompt: 


			'Direct play did NOT start up. Browser will not open.'.


		^self disconnectFromNetwork]. 


	super open.





close


	self disconnectFromNetwork.


	^super close





windowClose: w clientData: ignore1 callData: doitRec


	super windowClose: w clientData: ignore1 callData: doitRec.


	self disconnectFromNetwork





connectPollingProcess


	| pollingBlock |


	pollingBlock := [


		[true] whileTrue: [


			(Delay forMilliseconds: 


				self class pollingRateInMilliseconds) wait. 


			self directPlayInactive 


				ifTrue: [^self disconnectPollingProcess].


			[self receive] whileTrue: []]].


	pollingProcess := pollingBlock 


		forkAt: Processor activePriority





disconnectPollingProcess


	pollingProcess isNil ifTrue: [^self].


	pollingProcess terminate.


	pollingProcess := nil





network communication


connectToNetwork


	| pollingBlock |


	self openDirectPlay.


	self directPlayInactive ifTrue: [^self].


	self connectToProvider.


	self directPlayInactive ifTrue: [^self].


	(self booleanPrompt: 'Start a new network group?')


		ifTrue: [self connectToNewSession]


		ifFalse: [self connectToExistingSession].


	self directPlayInactive ifTrue: [^self].


	self connectPollingProcess





disconnectFromNetwork


	ActiveNetEnabledBrowser := nil.


	self disconnectPollingProcess.


	self closeDirectPlay





receive


	"Poll for a message and decode it; i.e., process it."


	“Returns true if a message exists; false, otherwise.”


	| strings selector |


	strings := directPlayObject receiveStrings.


	strings isEmpty ifTrue: [^false].





	selector := strings first asSymbol.


	(self respondsTo: selector) 


		ifFalse: [self directPlayGarbage: strings. ^false].


	self perform: selector with: strings.


	^true





send: aCollectionOfStrings


	directPlayObject sendStrings: aCollectionOfStrings





directPlayActive


	^directPlayObject notNil


directPlayInactive


	^directPlayObject isNil


isServer


	^directPlayObject logonType == #server





message processing


directPlayGarbage: strings


	Transcript cr; show: 'Garbage ignored: ', strings printString





directPlayHello: strings


	self isServer ifTrue: [self informDirectPlayOfSelection]. 





directPlaySelect: strings


	self privateSelectClass: (strings at: 2) selector: strings last.





message generation


informDirectPlayOfSelection


	| className methodName stream |


	className := self selectedClasses size = 1 


		ifTrue: [self selectedClasses first name] 


		ifFalse: [''].


	self showingInstanceMethods


		ifFalse: [className := className, ' class'].


	methodName := self selectedMethodNames size = 1 


		ifTrue: [self selectedMethodNames first]


		ifFalse: [''].


	self send: (Array with: #directPlaySelect: 


		with: className with: methodName)





privateSelectClass: classString selector: selectorString


	"If it exists, target the browser at the indicated method."


	self 


		withClassName: classString selectorName: selectorString 


		ifError:  [^self]


		otherwise: [:class :method :isInstanceMethod |


			self 


				classesSelected: (Array with: class);


				applications: (Array with: method application);


				selectedCategories: 


					(class categoriesFor: selectorString asSymbol);


				showingPublicClasses: method isPublic;


				showingInstanceMethods: isInstanceMethod;


				selectMethodNamed: method selector;


				selectClass: class;


				changedSelectedApplications.


			super changedSelectedMethods].





withClassName: classString selectorName: selectorString 


ifError: errorBlock otherwise: classAndMethodBlock


	"See privateSelectMethod:inClass for example usage."


	| hasClassSuffix className class dictionaryClass method |


	hasClassSuffix := classString size > 5 and: [


		(classString 


			copyFrom: classString size - 5 


			to: classString size) = ' class'].


	className := hasClassSuffix


		ifTrue: [classString copyFrom: 1 to: classString size - 6]


		ifFalse: [classString].


	class := Smalltalk at: className asSymbol 


		ifAbsent: [^errorBlock value].


	dictionaryClass := hasClassSuffix


		ifTrue: [class class]


		ifFalse: [class].


	method := dictionaryClass methodDictionary 


		at: selectorString asSymbol


		ifAbsent: [^errorBlock value].


	^classAndMethodBlock 


		value: class value: method value: hasClassSuffix not





overriding browser


changedSelectedMethods


	super changedSelectedMethods.


	self informDirectPlayOfSelection





prompters


booleanPrompt: title


	"Return a boolean."


	^System confirm: title


messagePrompt: title


	"Return nothing."


	^System message: title


stringPrompt: title


	"Return a string."


	^System prompt: title


selectFromList: aCollection title: title


	"Return an element of the collection."


	^CwListPrompter new


		parent: self shell;


		messageString: title;


		selectedItems: #();


		items: aCollection;


		printBlock: nil;


		prompt





support for network communication


openDirectPlay


	directPlayObject := DirectPlay new open: self guidString.


	self onErrorCloseDirectPlayWith: 


		'Failed to connect to network'





closeDirectPlay


	directPlayObject isNil ifFalse: [directPlayObject close].


	directPlayObject := nil





connectToProvider


	| list networkProvider networkGroup logonResult |


	list := directPlayObject networkProviders.


	networkProvider := 


		self selectFromList: list title: 'Choose a network provider.'.


	networkProvider isNil ifTrue: [^self close]. "user cancelled"


	directPlayObject chooseNetworkProvider: 


		(list indexOf: networkProvider).





connectToNewSession


	self logonNetwork: #logonAsServerNamed:and:.


	self send: #(directPlayHello:)





connectToExistingSession


	| list networkGroup message |


	list := directPlayObject networkSessions.


	list isEmpty ifTrue: [


		(self booleanPrompt: 


			'There are no existing sessions. Start a new one?')


			ifTrue: [^self connectToNewSession]


			ifFalse: [^self close]].


	networkGroup := self 


		selectFromList: list 


		title: 'Choose the network group to connect to.'.


	networkGroup isNil ifTrue: [^self]. "user cancelled"


	directPlayObject chooseSession: 


		(list indexOf: networkGroup).


	self logonNetwork: #logonAsClientNamed:and:.


	self directPlayInactive ifTrue: [^self].


	self send: #(directPlayHello:).


	[self receive] whileFalse: [].





logonNetwork: logonType


	| identification separatorIndex friendlyName formalName |


	identification := 


		self stringPrompt: 'Formal name, friendly name'.


	separatorIndex := identification indexOf: $,.


	separatorIndex = 0 


		ifTrue: [formalName := friendlyName := identification]


		ifFalse: [


			formalName := 


				identification copyFrom: 1 to: separatorIndex - 1. 


			friendlyName := 


				identification copyFrom: separatorIndex + 1 to: 


					identification size].


	directPlayObject 


		perform: logonType 


		with: formalName trimBlanks 


		with: friendlyName trimBlanks.


	self onErrorCloseDirectPlayWith: 'Failed to logon'





onErrorCloseDirectPlayWith: message


	self error ifTrue: [


		self messagePrompt: message, ': ', self errorString.


		self disconnectFromNetwork]





error


	^self errorString notEmpty


errorString


	^directPlayObject errorString





guidString


	"This global unique identifier was obtained by running 


	c:\msdev\bin\uuidgen.exe in Microsoft Windows 95."


	^'{d4ba6b60-00d7-11d0-88d2-00a024a57dd9}' asDBString

